
Abstract. This work is a case study in program verification: We have
written a simple parser and a corresponding pretty-printer in a non-
strict functional programming language with lifted pairs and functions
(Haskell). A natural aim is to prove that the programs are, in some
sense, each other’s inverses. The presence of partial and infinite values
in the domains makes this exercise interesting, and having lifted types
adds an extra spice to the task. We have tackled the problem in differ-
ent ways, and this is a report on the merits of those approaches. More
specifically, we first describe a method for testing properties of programs
in the presence of partial and infinite values. By testing before proving
we avoid wasting time trying to prove statements that are not valid.
Then we prove that the programs we have written are in fact (more or
less) inverses using first fixpoint induction and then the approximation
lemma.

1 Introduction

Infinite values are commonly used in (non-strict) functional programs, often
to improve modularity [5]. Partial values are seldom used explicitly, but they
are still present in all non-trivial Haskell programs because of non-termination,
pattern match failures, calls to the error function etc. Unfortunately, proofs
about functional programs often ignore details related to partial and infinite
values.

This text is a case study where we explore how one can go about testing
and proving properties even in the presence of partial and infinite values. We
use random testing (Sect. 5) and two proof methods, fixpoint induction (Sect. 7)
and the approximation lemma (Sect. 8), both described in Gibbons’ and Hutton’s
tutorial [4].

The programs that our case study revolves around are a simple pretty-printer
and a corresponding parser. Jansson and Jeuring define several more complex
(polytypic) pretty-printers and parsers and prove them correct for total, finite
input [7]. The case study in this paper uses cut down versions of those programs

(see Sect. 2) but proves a stronger statement. On some occasions we have been
tempted to change the definitions of the programs to be able to formulate our
proofs in a different way. We have not done that, since one part of our goal is
to explore what it is like to prove properties about programs that have not been
written with a proof in mind. We have transformed our programs into equivalent
variants, though; note that this carries a proof obligation.

Before starting to prove something it is often useful to test the properties.
That way one can avoid spending time trying to prove something which is not
true anyway. However, testing partial and infinite values can be tricky. In Sect. 5
we describe two techniques for doing that. Infinite values can be tested with the
aid of the approximation lemma, and for partial values we make use of a Haskell
extension, implemented in several Haskell environments. (The first technique is
a generalisation of another one, and the last technique is previously known.)

As indicated above the programming language used for all programs and
properties is Haskell [12], a non-strict, pure functional language where all types
are lifted. Since we are careful with all details there will necessarily be some
Haskell-specific discussions below, but the main ideas should carry over to other
similar languages. Some knowledge of Haskell is assumed of the reader, though.

We begin in Sect. 2 by defining the two programs that this case study focuses
on. Section 3 discusses the computational model and in Sect. 4 we give idealised
versions of the main properties that we want to prove. By implementing and
testing the properties (in Sect. 5) we identify a flaw in one of the them and we
give a new, refined version in Sect. 6. The proofs presented in Sects. 7 and 8 are
discussed in the concluding Sect. 9.

2 Programs

The programs under consideration parse and pretty-print a simple binary tree
data type T without any information in the nodes:

data T = L | B T T

The pretty-printer is really simple. It performs a preorder traversal of the
tree, emitting a ’B’ for each branching point and an ’L’ for each leaf:

pretty ′ :: T → String
pretty ′ L = "L"
pretty ′ (B l r) = "B" ++ pretty ′ l ++ pretty ′ r

The parser reconstructs a tree given a string of the kind produced by pretty ′.
Any remaining input is returned together with the tree:

parse :: String → (T ,String)
parse (’L’ : cs) = (L, cs)
parse (’B’ : cs) = (B l r , cs ′′)

where (l , cs ′) = parse cs
(r , cs ′′) = parse cs ′



We wrap up pretty ′ so that the printer and the parser get symmetric types:

pretty :: (T ,String) → String
pretty (t , cs) = pretty ′ t ++ cs

These programs are obviously written in a very naive way. A real pretty-
printer would not use a quadratic algorithm for printing trees and a real parser
would use a proper mechanism for reporting parse failures. However, the pro-
grams have the right level of detail for our application; they are very straightfor-
ward without being trivial. The tree structure makes the recursion “nonlinear,”
and that is what makes these programs interesting.

3 Computational Model

Before we begin reasoning about the programs we should specify what our un-
derlying computational model is. We use Haskell 98 [12], and it is common to
reason about Haskell programs by using equational reasoning, assuming that a
simple denotational semantics for the language exists. This is risky, though, since
this method has not been formally verified to work; there is not even a formal
semantics for the language to verify it against. (We should mention that some
work has been done on the static semantics [3].)

Nevertheless we will follow this approach, taking some caveats into account
(see below). Although our aim is to explore what a proof would look like when all
issues related to partial and infinite values are considered, it may be that we have
missed some subtle aspect of the Haskell semantics. We have experimented with
different levels of detail and believe that the resolution of such issues most likely
will not change the overall structure of the proofs, though. Even if we would
reject the idea of a clean denotational semantics for Haskell and instead use
Sands’ improvement theory [13] based on an operational model, we still believe
that the proof steps would be essentially the same.

Now on to the caveats. All types in Haskell are (by default) pointed and lifted;
each type is a complete partial order with a distinct least element ⊥ (bottom),
and data constructors are not strict. For pairs this means that ⊥ 6= (⊥,⊥), so
we do not have surjective pairing. It is possible to use strictness annotations
to construct types that are not lifted, e.g. the smash product of two types, for
which ⊥ = (⊥,⊥) but we still do not have surjective pairing. There is however
no way to construct the ordinary cartesian product of two types.

One has to be careful when using pattern matching in conjunction with lifted
types. The expression let (a, b) = x in g (a, b) is equivalent to g x iff x 6= ⊥
or g (⊥,⊥) = g ⊥. The reason is that, if x = ⊥, then in the first case g will
still be applied to (⊥,⊥), whereas in the second case g will be applied to ⊥.
Note here that the pattern matching in a let clause is not performed until the
variables bound in the pattern are actually used. Hence let (a, b) = ⊥ in (a, b)
is equivalent to (⊥,⊥), whereas (λ(a, b) → (a, b)) ⊥ = ⊥.

The function type is also lifted; we can actually distinguish between ⊥::a → a
and λx → ⊥ :: a → a by using seq , a function with the following semantics [12]:



seq :: a → b → b
seq ⊥ b = ⊥
seq a b = b

(Here a is any value except for ⊥.) In other words η-conversion is not valid for
Haskell functions, so to verify that two functions are equal it is not enough to
verify that they produce identical output when applied to identical input; we also
have to verify that none (or both) of the functions are ⊥. A consequence of the
lack of η-conversion is that one of the monadic identity laws fails to hold for some
standard Monad instances in Haskell, such as the state “monad.” The existence
of a polymorphic seq also weakens Haskell’s parametricity properties [8], but
that does not directly affect us because our functions are not polymorphic.

Another caveat, also related to seq , is that f = λTrue x → x is not identical
to f ′ = λTrue → λx → x . By careful inspection of Haskell’s pattern matching
semantics [12] we can see that f False = λx → ⊥ while f ′ False = ⊥, since the
function f is interpreted as

λa → λb → case (a, b) of
(True, x ) → x

whereas the function f ′ is interpreted as

λa → case a of
True → λx → x .

This also applies if f and f ′ are defined by f True x = x and f ′ True = λx → x .
We do not get any problems if the first pattern is a simple variable, though. We
will avoid problems related to this issue by never pattern matching on anything
but the last variable in a multiple parameter function definition.

4 Properties: First Try

The programs in Sect. 2 are simple enough. Are they correct? That depends on
what we demand of them. Let us say that we want them to form an embedding-
projection pair, i.e.

parse ◦ pretty = id :: (T ,String) → (T ,String) (1)

and

pretty ◦ parse v id :: String → String . (2)

The operator v denotes the ordering of the semantical domain, and = is seman-
tical equality.

More concretely (1) means that for all pairs p :: (T ,String) we must have
parse (pretty p) = p. (Note that η-conversion is valid since none of the functions
involved are equal to ⊥; they both expect at least one argument.) The quantifi-
cation is over all pairs of the proper type, including infinite and partial values.



If we can prove this equality, then we are free to exchange the left and right
hand sides in any well-typed context. This means that we can use the result
very easily, but we have to pay a price in the complexity of the proof. In this
section we “cheat” by only quantifying over finite, total trees so that we can use
simple structural induction. We return to the full quantification in later sections.

Parse after Pretty. Let us prove (1) for finite, total trees and arbitrary strings,
just to illustrate what this kind of proof usually looks like. First we observe that
both sides are distinct from ⊥, and then we continue using structural induction.
The inductive hypothesis used is

∀cs :: String . (parse ◦ pretty) (t , cs) = id (t , cs),

where t :: T is any immediate subtree of the tree treated in the current case.
We have two cases, for the two constructors of T . The first case is easy (for an
arbitrary cs :: String):

(parse ◦ pretty) (L, cs)

= {◦}
parse (pretty (L, cs))

= {pretty}
parse (pretty ′ L ++ cs)

= {pretty ′}
parse ("L" ++ cs)

= {++}
parse (’L’ : cs)

= {parse}
(L, cs)

The second case requires somewhat more work, but is still straightforward.
(The use of where here is not syntactically correct, but is used for stylistic
reasons. Just think of it as a postfix let.)

(parse ◦ pretty) (B l r , cs)

= {◦, pretty}
parse (pretty ′ (B l r) ++ cs)

= {pretty ′, ++ associative, ++}
parse (’B’ : pretty ′ l ++ pretty ′ r ++ cs)

= {parse}
(B l ′ r ′, cs ′′)

where (l ′, cs ′) = parse (pretty ′ l ++ pretty ′ r ++ cs)
(r ′, cs ′′) = parse cs ′

= {pretty , ◦}
(B l ′ r ′, cs ′′)

where (l ′, cs ′) = (parse ◦ pretty) (l , pretty ′ r ++ cs)
(r ′, cs ′′) = parse cs ′

= {Inductive hypothesis}



(B l ′ r ′, cs ′′)
where (l ′, cs ′) = id (l , pretty ′ r ++ cs)

(r ′, cs ′′) = parse cs ′

= {id , where}
(B l r ′, cs ′′)

where (r ′, cs ′′) = parse (pretty ′ r ++ cs)
= {pretty , ◦}
(B l r ′, cs ′′)

where (r ′, cs ′′) = (parse ◦ pretty) (r , cs)
= {Inductive hypothesis}
(B l r ′, cs ′′)

where (r ′, cs ′′) = id (r , cs)
= {id , where}
(B l r , cs)

Hence we have proved using structural induction that (parse ◦ pretty) (t , cs)
= (t , cs) for all finite, total t :: T and for all cs :: String . Thus we can draw the
conclusion that (1) is satisfied for that kind of input.

Pretty after Parse. We can show that (2) is satisfied in a similar way, using the
fact that all Haskell functions are continuous and hence monotone with respect
to v. In fact, the proof works for arbitrary partial, finite input. We show the
case for cs :: String , head cs = ’B’, i.e. cs = ’B’ : cs1 for some (partial and
finite) cs1 :: String :

(pretty ◦ parse) (’B’ : cs1)
= {◦, parse}
pretty (B l r , cs ′′1 )

where (l , cs ′1) = parse cs1

(r , cs ′′1 ) = parse cs ′1
= {pretty , pretty ′, ++ associative}
"B" ++ pretty ′ l ++ pretty ′ r ++ cs ′′1

where (l , cs ′1) = parse cs1

(r , cs ′′1 ) = parse cs ′1
= {pretty}
"B" ++ pretty ′ l ++ pretty (r , cs ′′1 )

where (l , cs ′1) = parse cs1

(r , cs ′′1 ) = parse cs ′1
= {where, pretty ⊥ = pretty (⊥,⊥), ◦}
"B" ++ pretty ′ l ++ (pretty ◦ parse) cs ′1

where (l , cs ′1) = parse cs1

v {Inductive hypothesis, monotonicity}
"B" ++ pretty ′ l ++ id cs ′1

where (l , cs ′1) = parse cs1

= {id , pretty , where, pretty ⊥ = pretty (⊥,⊥), ◦}
"B" ++ (pretty ◦ parse) cs1

v {Inductive hypothesis, monotonicity}
"B" ++ id cs1

= {id , ++}
’B’ : cs1

The other cases (head cs 6∈ {’L’, ’B’} and head cs = ’L’) are both straightfor-
ward.



Parse after Pretty, Revisited. If we try to allow partial input in (1) instead
of only total input, then we run into problems, as this counterexample shows:

(parse ◦ pretty) (⊥, cs)
= {◦, pretty}
parse (pretty ′ ⊥++ cs)

= {pretty ′, ++}
parse ⊥

= {parse}
⊥ :: (T ,String)

6= {(, ) is not strict}
(⊥, cs) :: (T ,String)

We summarise our results so far in a table; we have proved (2) for finite,
partial input and (1) for finite, total input. We have also disproved (1) in the
case of partial input. The case marked with ? is treated in Sect. 5 below.

Total Partial
Finite (2), (1) (2), ¬ (1)
Infinite ? ¬ (1)

Hence the programs are not correct if we take (1) and (2) plus the type signatures
of pretty and parse as our specification. Instead of refining the programs to meet
this specification we will try to refine the specification. This approach is in line
with our goal from Sect. 1: To prove properties of programs, without changing
them.

5 Tests

As seen above we have to refine our properties, at least (1). To aid us in finding
properties which are valid for partial and infinite input we will test the properties
before we try to prove them.

How do we test infinite input in finite time? An approach which seems to
work fine is to use the approximation lemma [6]. For T the function approx is
defined as follows (Nat is a data type for natural numbers):

data Nat = Zero | Succ Nat
approx :: Nat → T → T
approx (Succ n) = λt → case t of

L → L
B l r → B (approx n l) (approx n r)

Note that approx Zero is undefined, i.e. ⊥. Hence approx n t traverses n levels
down into the tree t and replaces everything there by ⊥.

For the special case of trees the approximation lemma states that, for any
t1, t2 :: T ,

t1 = t2 iff ∀n ∈ Natfin . approx n t1 = approx n t2. (3)



Here Natfin stands for the total and finite values of type Nat , i.e. Natfin corre-
sponds directly to N. If we want to test that two expressions yielding possibly
infinite trees are equal then we can use the right hand side of this equivalence.
Of course we cannot test the equality for all n, but if it is not valid then running
the test for small values of n should often be enough to find a counterexample.

Testing equality between lists using take :: Int → [a ] → [a ] and the take
lemma, an analogue to the approximation lemma, is relatively common. However,
the former does not generalise as easily to other data types as the latter does. The
approximation lemma generalises to any type which can be defined as the least
fixpoint of a locally continuous functor [6]. This includes not only all polynomial
types, but also much more, like nested and exponential types.

Using the approximation lemma we have now reduced testing of infinite val-
ues to testing of partial values. Thus even if we were dealing with total values
only, we would still need to include ⊥ in our tests. Generating the value ⊥ is
easily accomplished:

⊥ :: a
⊥ = error "_|_"

(Note that the same notation is used for the expression that generates a ⊥ as
for the value itself.)

The tricky part is testing for equality. If we do not want to use a separate
tool then we necessarily have to use some impure extension, e.g. exception han-
dling [11]. Furthermore it would be nice if we could perform these tests in pure
code, such as QuickCheck [2] properties (see below). This can only be accom-
plished by using the decidedly unsafe function unsafePerformIO :: IO a → a
[1, 11]. The resulting function isBottom :: a → Bool1 has to be used with care;
it only detects a ⊥ that results in an exception. However, that is enough for
our purposes, since pattern match failures, error "..." and undefined all raise
exceptions. If isBottom x terminates properly, then we can be certain that the
answer produced (True or False) is correct.

Using isBottom we define a function that compares two arbitrary finite trees
for equality:

(=̂) :: T → T → Bool
t1 =̂ t2 = case (isBottom t1, isBottom t2) of

(True,True) → True
(False,False) → case (t1, t2) of

(L,L) → True
(B l r ,B l ′ r ′) → l =̂ l ′ ∧ r =̂ r ′

→ False
→ False

1 The function isBottom used here is a slight variation on the version implemented by
Andy Gill in the libraries shipped with the GHC Haskell compiler. We have to take
care not to catch e.g. stack overflow exceptions, as these may or may not correspond
to bottoms.



Similarly we can define a function (v̂) :: T → T → Bool which implements
an approximation of the semantical domain ordering (v). The functions approx ,
(=̂) and (v̂) are prime candidates for generalisation. We have implemented them
using type classes; instances are generated automatically using the “Scrap Your
Boilerplate” approach to generic programming [9].

QuickCheck is a library for defining and testing properties of Haskell func-
tions [2]. By using the framework developed above we can now give QuickCheck
implementations of properties (1) and (2):

prop1 n = forAll pair (λp →
approxPair n ((parse ◦ pretty) p) =̂ approxPair n (id p))

prop2 n = forAll string (λcs →
approx n ((pretty ◦ parse) cs) v̂ approx n (id cs))

approxPair n (t , cs) = (approx n t , approx (2ˆn) cs)

These properties can be read more or less as ordinary set theoretic predicates,
e.g. for prop1 “for all pairs p the equality . . . holds.” The generators pair and
string (defined in Appendix A) ensure that many different finite and infinite
partial values are used for p and cs in the tests. Some values are never generated,
though; see the end of this section.

If we run these tests then we see that prop1 fails almost immediately, whereas
prop2 succeeds all the time. In other words (1) is not satisfied (which we already
knew, see Sect. 4), but on the other hand we can be relatively certain that (2)
is valid.

You might be interested in knowing whether (1) holds for total infinite input,
a case which we have neglected above. We can easily write a test for such a case:

infiniteTree = B infiniteTree L
propInfiniteTotal n =

approxPair n ((parse ◦ pretty) p) =̂ approxPair n (id p)
where p = (infiniteTree, "")

(The value infiniteTree is a left-infinite tree.) When executing this test we run
into trouble, though; the test does not terminate for any n ∈ Natfin. The reason is
that the left-hand side does not terminate, and no part of the second component
of the output pair is ever created (i.e. it is ⊥). This can be seen by unfolding the
expression a few steps:

approxPair n ((parse ◦ pretty) (infiniteTree, ""))
= {Unfold, rearrange slightly}
(approx n (B l r), approx (2ˆn) cs ′′)

where (l , cs ′) = (parse ◦ pretty) (infiniteTree, "L")
(r , cs ′′) = parse cs ′

One of the subexpressions is (parse ◦ pretty) (infiniteTree, "L"), which is essen-
tially the same expression as the one that we started out with, and cs ′′ will not
be generated until that subexpression has produced any output in its second

Fig. 1: With the left tree called t , the right tree is t ′ = (fst ◦ parse ◦ pretty ′) t .

component. The right-hand side does terminate, though, so (1) is not valid for
total, infinite input.

Since prop1 does not terminate for total, infinite trees we have designed our
QuickCheck generators so that they do not generate such values. This is of course
a slight drawback.

6 Properties: Second Try

As noted above (1) is not valid in general. If we inspect what happens when
fst ◦ parse ◦ pretty ′ is applied to a partial tree, then we see that as soon as a ⊥
is encountered all nodes encountered later in a preorder traversal of the tree are
replaced by ⊥ (see Fig. 1).

We can easily verify that the example in the figure is correct (assuming that
the part represented by the vertical dots is a left-infinite total tree):

t = B (B (B ⊥ infiniteTree) L ) (B L L)
t ′ = B (B (B ⊥ ⊥ ) ⊥) ⊥
propFigure = t ′ =̂ (fst ◦ parse ◦ pretty ′) t

Evaluating propFigure yields True, as expected.
Given this background it is not hard to see that (snd ◦parse ◦pretty) (t , cs) =

⊥ whenever the tree t is not total. Furthermore (parse ◦ pretty) (t , cs) = ⊥ iff
t = ⊥. Using the preceding results we can write a replacement strictify for id
that makes

parse ◦ pretty = strictify :: (T ,String) → (T ,String) (1’)

a valid refinement of (1) (as we will see below):

strictify :: (T , a) → (T , a)
strictify (t , a) = t ‘seq ‘ (t ′, tTotal ‘seq ‘ a)

where (t ′, tTotal) = strictify ′ t



If t = ⊥ then ⊥ should be returned, hence the first seq . The helper function
strictify ′, which does the main trunk of the work, returns the strictified tree in its
first component. The second component, which is threaded bottom-up through
the computation, is () whenever the input tree is total, and ⊥ otherwise; hence
the second seq . In effect we use the Haskell type () as a boolean type with ⊥
as falsity and () as truth. It is the threading of this “boolean,” in conjunction
with the sequential nature of seq , which enforces the preorder traversal and
strictification indicated in the figure above:

strictify ′ :: T → (T , ())
strictify ′ L = (L, ())
strictify ′ (B l r) = (B l ′ (lTotal ‘seq ‘ r ′), lTotal ‘seq ‘ rTotal)

where (l ′, lTotal) = strictify ′ l
(r ′, rTotal) = strictify ′ r

Note that if the left subtree l is not total, then the right subtree r should be
replaced by ⊥; hence the use of lTotal ‘seq ‘ r ′ above. The second component
should be () iff both subtrees are total, so we use seq as logical and between
lTotal and rTotal ; a ‘seq ‘ b = () iff a = () and b = () for a, b :: ().

Before we go on to prove (1’), let us test it:

prop′1 n = forAll pair (λp →
approxPair n ((parse ◦ pretty) p) =̂ approxPair n (strictify p))

This test seems to succeed all the time — a good indication that we are on the
right track.

7 Proofs Using Fixpoint Induction

Now we will prove (1’) and (2) using two different methods, fixpoint induction
(in this section) and the approximation lemma (in Sect. 8). All details will not
be presented, since that would take up too much space.

In this section let ψ, ψi etc. stand for arbitrary types.
To be able to use fixpoint induction [4, 14] all recursive functions have to be

defined using fix , which is defined by

fixf =
∞⊔

i=0

f i⊥ (4)

for any continuous function f ::ψ → ψ. (The notation f i stands for f composed
with itself i times.) It is easy to implement fix in Haskell, but proving that the
two definitions are equivalent would take up too much space, and is omitted:

fix :: (a → a) → a
fix f = f (fix f )



Let P be a chain-complete predicate, i.e. a predicate which is true for the
least upper bound of a chain whenever it is true for all the elements in the chain.
In other words, if P (f i⊥) is true for all i ∈ N and some f :: ψ → ψ, then we
know that P (fix f) is true (we only consider ω-chains). Generalising we get the
following inference rule from ordinary induction over natural numbers (and some
simple domain theory):

P (⊥,⊥, . . . ,⊥)
∀n ∈ N . (P (fn

1 ⊥, fn
2 ⊥, . . . , fn

m⊥) ⇒
P (fn+1

1 ⊥, fn+1
2 ⊥, . . . , fn+1

m ⊥))
P (fix f1,fix f2, . . . ,fix fm) (5)

Here m ∈ N and the fi are continuous functions fi :: ψi → ψi. We also have
the following useful variant which follows immediately from the previous one,
assuming that the ψi are function types, ψi = ψ′i → ψ′′i , and that all fi are
strictness-preserving, i.e. if gi is strict then fi gi should be strict as well.

P (⊥,⊥, . . . ,⊥)
∀ strict g1 :: ψ1, g2 :: ψ2, . . . , gm :: ψm .

P (g1, g2, . . . , gm) ⇒ P (f1 g1, f2 g2, . . . , fm gm)
P (fix f1,fix f2, . . . ,fix fm) (6)

That is all the theory that we need for now; on to the proofs. Let us begin
by defining variants of our recursive functions using fix :

pretty ′fix :: T → String
pretty ′fix = fix prettystep
prettystep :: (T → String) → T → String
prettystep p L = "L"

prettystep p (B l r) = "B" ++ p l ++ p r

parsefix :: String → (T ,String)
parsefix = fix parsestep
parsestep :: (String → (T ,String)) → String → (T ,String)
parsestep p′ (’L’ : cs) = (L, cs)
parsestep p′ (’B’ : cs) = (B l r , cs ′′)

where (l , cs ′) = p′ cs
(r , cs ′′) = p′ cs ′

strictify ′fix :: T → (T , ())
strictify ′fix = fix strictifystep
strictifystep :: (T → (T , ())) → T → (T , ())
strictifystep s L = (L, ())
strictifystep s (B l r) = (B l ′ (lTotal ‘seq ‘ r ′), lTotal ‘seq ‘ rTotal)

where (l ′, lTotal) = s l
(r ′, rTotal) = s r



Of course using these definitions instead of the original ones implies a proof
obligation; we have to show that the two sets of definitions are equivalent to
each other. In a standard domain theoretic setting this would follow immediately
from the interpretation of a recursively defined function. In the case of Haskell
this requires some work, though. The proofs are certainly possible to perform,
but they would lead us too far astray, so we omit them here.

The properties have to be unrolled to fit the requirements of the inference
rules. To make the properties more readable we define new versions of some other
functions as well:

prettyfix :: (T → String) → (T ,String) → String
prettyfix p (t , cs) = p t ++ cs

strictifyfix :: (T → (T , ())) → (T , a) → (T , a)
strictifyfix s (t , a) = t ‘seq ‘ (t ′, tTotal ‘seq ‘ a)

where (t ′, tTotal) = s t

We end up with

P1(p, p′, s) = (7)
p′ ◦ prettyfix p = strictifyfix s

and

P2(p, p′) = (8)
prettyfix p ◦ p′ v id .

However, we cannot use P1 as it stands since P1(⊥,⊥,⊥) is not true. To see this,
pick an arbitrary cs :: String and a t :: T satisfying t 6= ⊥:

(⊥ ◦ prettyfix ⊥) (t , cs)
= {◦, ⊥}
⊥ :: (T ,String)

6= {seq , t 6= ⊥, (, ) is not strict}
t ‘seq ‘ (⊥,⊥) :: (T ,String)

= {seq}
t ‘seq ‘ (⊥,⊥ ‘seq ‘ cs) :: (T ,String)

= {where, pattern matching}
t ‘seq ‘ (t ′, tTotal ‘seq ‘ cs) :: (T ,String)

where (t ′, tTotal) = ⊥
= {⊥}
t ‘seq ‘ (t ′, tTotal ‘seq ‘ cs) :: (T ,String)

where (t ′, tTotal) = ⊥ t
=

˘
strictifyfix

¯
strictifyfix ⊥ (t , cs)

We can still go on by noticing that we are only interested in the property in
the limit and redefining it as

P ′1(p, p
′, s) = P1(prettystep p, parsestep p′, strictifystep s), (7’)



i.e. P ′1(p, p
′, s) is equivalent to

parsestep p′ ◦ prettyfix (prettystep p) = strictifyfix (strictifystep s). (9)

With P ′1 we avoid the troublesome base case since P ′1(⊥,⊥,⊥) is equivalent to
P1(prettystep ⊥, parsestep ⊥, strictifystep ⊥).

Now it is straightforward to verify that P ′1(⊥,⊥,⊥) and P2(⊥,⊥) are valid
(P ′1 requires a tedious but straightforward case analysis). It is also easy to
verify that the predicates are chain-complete using general results from do-
main theory [14]. As we have already stated above, verifying formally that
P ′1(fix prettystep , fix parsestep , fix strictifystep) is equivalent to (1’) and simi-
larly that P2(fix prettystep , fix parsestep) is equivalent to (2) requires more work
and is omitted.

Pretty after Parse. Now on to the main work. Let us begin with P2. Since
we do not need the tighter inductive hypothesis of inference rule (5) we will use
inference rule (6); it is easy to verify that prettystep and parsestep are strictness-
preserving. Assume now that P2(p, p′) is valid for strict p :: T → String and
p′ :: String → (T ,String). We have to show that P2(prettystep p, parsestep p′) is
valid. After noting that both sides of the inequality are distinct from ⊥, take an
arbitrary element cs :: String . The proof is a case analysis on head cs.

First case, head cs 6∈ {’L’, ’B’}:

(prettyfix (prettystep p) ◦ (parsestep p′)) cs

=
˘
◦, parsestep , head cs 6∈ {’L’, ’B’}

¯
prettyfix (prettystep p) ⊥

=
˘
prettyfix

¯
⊥ :: String

v {⊥ is the least element in the domain}
id cs

Second case, head cs = ’L’, i.e. cs = ’L’ : cs1 for some cs1 :: String :

(prettyfix (prettystep p) ◦ (parsestep p′)) cs

=
˘
◦, parsestep , cs = ’L’ : cs1

¯
prettyfix (prettystep p) (L, cs1)

=
˘
prettyfix

¯
prettystep p L ++ cs1

=
˘
prettystep , ++, id

¯
id cs

Last case, head cs = ’B’, i.e. cs = ’B’ : cs1 for some cs1 :: String :

(prettyfix (prettystep p) ◦ (parsestep p′)) cs

=
˘
◦, parsestep , cs = ’B’ : cs1

¯
prettyfix (prettystep p) (B l r , cs ′′1 )

where (l , cs ′1) = p′ cs1

(r , cs ′′1 ) = p′ cs ′1



=
˘
prettyfix , prettystep , ++ associative

¯
"B" ++ p l ++ (p r ++ cs ′′1 )

where (l , cs ′1) = p′ cs1

(r , cs ′′1 ) = p′ cs ′1
=

˘
prettyfix

¯
"B" ++ p l ++ prettyfix p (r , cs ′′1 )

where (l , cs ′1) = p′ cs1

(r , cs ′′1 ) = p′ cs ′1
=

˘
where, p strict implies that prettyfix p ⊥ = prettyfix p (⊥,⊥), ◦

¯
"B" ++ p l ++ (prettyfix p ◦ p′) cs ′1

where (l , cs ′1) = p′ cs1

v {Inductive hypothesis, monotonicity}
"B" ++ p l ++ id cs ′1

where (l , cs ′1) = p′ cs1

=
˘
id , prettyfix , where, p strict, ◦

¯
"B" ++ (prettyfix p ◦ p′) cs1

v {Inductive hypothesis, monotonicity}
"B" ++ id cs1

= {id , ++, id}
id cs

This concludes the proof for P2.

Parse after Pretty. For P ′1 we will also use inference rule (6); in addition
to prettystep and parsestep it is easy to verify that strictifystep is strictness-
preserving. Assume that P ′1(p0, p′0, s0) is valid, where p0, p′0 and s0 are all strict.
We have to prove that P ′1(prettystep p0, parsestep p′0, strictifystep s0) is valid.
This is equivalent to proving P1(prettystep p, parsestep p′, strictifystep s), where
p = prettystep p0, p′ = parsestep p′0 and s = strictifystep s0. The first step of
this proof is to note that both sides of the equality in P ′1 are distinct from ⊥.
The rest of the proof is, as before, performed using case analysis, this time on
pair , an arbitrary element in (T , cs). The cases pair = ⊥, pair = (⊥, cs) and
pair = (L, cs) for an arbitrary cs :: String are straightforward and omitted.

Last case, pair = (B l r , cs) for arbitrary subtrees l , r :: T and an arbitrary
cs :: String :

(parsestep p′ ◦ prettyfix (prettystep p)) (B l r , cs)

=
˘
◦, prettyfix

¯
parsestep p′ (prettystep p (B l r) ++ cs)

=
˘
prettystep , ++, ++ associative

¯
parsestep p′ (’B’ : p l ++ (p r ++ cs))

=
˘
parsestep

¯
(B l ′ r ′, cs ′′)

where (l ′, cs ′) = p′ (p l ++ (p r ++ cs))
(r ′, cs ′′) = p′ cs ′

=
˘
prettyfix , ◦

¯



(B l ′ r ′, cs ′′)
where (l ′, cs ′) = (p′ ◦ prettyfix p) (l , p r ++ cs)

(r ′, cs ′′) = p′ cs ′

= {Inductive hypothesis}
(B l ′ r ′, cs ′′)

where (l ′, cs ′) = strictifyfix s (l , p r ++ cs)

(r ′, cs ′′) = p′ cs ′

=
˘
strictifyfix

¯
(B l ′ r ′, cs ′′)

where (l ′, cs ′) = l ‘seq ‘ (t ′, tTotal ‘seq ‘ p r ++ cs)
(t ′, tTotal) = s l
(r ′, cs ′′) = p′ cs ′

= {Simple case analysis on l (⊥ or not ⊥), pattern matching}
(B l ′ r ′, cs ′′)

where (l ′, cs ′) = (l ‘seq ‘ t ′, l ‘seq ‘ tTotal ‘seq ‘ p r ++ cs)
(t ′, tTotal) = s l
(r ′, cs ′′) = p′ cs ′

= {seq , if l = ⊥ then t ′ = tTotal = ⊥ since s is strict}
(B l ′ r ′, cs ′′)

where (l ′, cs ′) = (t ′, tTotal ‘seq ‘ p r ++ cs)
(t ′, tTotal) = s l
(r ′, cs ′′) = p′ cs ′

= {where}
(B t ′ r ′, cs ′′)

where (t ′, tTotal) = s l
(r ′, cs ′′) = p′ (tTotal ‘seq ‘ p r ++ cs)

= {Rename variables}
(B l ′ r ′, cs ′′)

where (l ′, lTotal) = s l
(r ′, cs ′′) = p′ (lTotal ‘seq ‘ p r ++ cs)

The rest of the proof is straightforward. Using case analysis on lTotal we
prove that

(B l ′ r ′, cs ′′)
where (r ′, cs ′′) = p′ (lTotal ‘seq ‘ p r ++ cs)

=
(B l ′ (lTotal ‘seq ‘ r ′), lTotal ‘seq ‘ rTotal ‘seq ‘ cs)

where (r ′, rTotal) = s r

is valid. In one branch one can observe that p′ is strict. In the other the inductive
hypothesis can be applied, followed by reasoning analogous to the one for l ‘seq ‘
above. Given this equality the rest of the proof is easy. Hence we can draw the
conclusion that P1(prettystep p, parsestep p′, strictifystep s) is valid, which means
that we have finished the proof.

8 Proofs Using the Approximation Lemma

Let us now turn to the approximation lemma. This lemma was presented above
in Sect. 5, but we still have a little work to do before we can go to the proofs.



Pretty after Parse. Any naive attempt to prove (2) using the obvious inductive
hypothesis fails. Using the following less obvious reformulated property does the
trick, though:

∀m ∈ N . ppm v id :: String → String . (10)

Here we use a family of helper functions ppm (m ∈ N):

ppm cs = pretty ′ t1 ++ pretty ′ t2 ++ . . .++ pretty ′ tm ++ csm

where (t1, cs1) = parse cs
(t2, cs2) = parse cs1

...
(tm, csm) = parse csm−1

(We interpret pp0 as id .) It is straightforward to verify that this property is
equivalent to (2).

Note that we cannot use the approximation lemma directly as it stands,
since the lemma deals with equalities, not inequalities. However, replacing each
= with v in the proof of the approximation lemma in Gibbons’ and Hutton’s
article [4, Sect. 4] is enough to verify this variant. We get that, for all m ∈ N
and cs :: String ,

ppm cs v id cs iff
∀n ∈ Natfin . approx n (ppm cs) v approx n (id cs).

(11)

Hence all that we need to do is to prove the last statement above (after noticing
that both ppm and id are distinct from ⊥, for all m ∈ N). We do that by
induction over n, after observing that we can change the order of the universal
quantifiers so that we get

∀n ∈ Natfin . ∀m ∈ N . ∀cs :: String .
approx n (ppm cs) v approx n (id cs), (12)

which is equivalent to the inequalities above.
For lists we have the following variant of approx :

approx :: Nat → [a ] → [a ]
approx (Succ n) = λ(x : xs) → x : approx n xs

Since approx Zero is undefined the statement (12) is trivially true for n = Zero.
Assume now that ∀m ∈ N . ∀cs ::String . approx n (ppm cs) v approx n (id cs)
is true for some n ∈ Natfin. Take an arbitrary m ∈ N. Note that the property
that we want to prove is trivially true for m = 0, so assume that m ≥ 1. We
proceed by case analysis on head cs.

First case, head cs 6∈ {’L’, ’B’}:
approx (Succ n) (ppm cs)

= {parse, where, pretty ′, ++}
approx (Succ n) ⊥

v {⊥ is the least element, monotonicity}
approx (Succ n) (id cs)



Second case, head cs = ’L’, i.e. cs = ’L’ : cs ′ for some cs ′ :: String :

approx (Succ n) (ppm (’L’ : cs ′))
= {ppm, m ≥ 1}
approx (Succ n) (pretty ′ t1 ++ pretty ′ t2 ++ . . . ++ pretty ′ tm ++ csm)

where (t1, cs1) = parse (’L’ : cs ′)
(t2, cs2) = parse cs1

...
(tm, csm) = parse csm−1

= {parse, where, note that if m = 1 then csm = cs ′}
approx (Succ n) (pretty ′ L ++ pretty ′ t2 ++ . . . ++ pretty ′ tm ++ csm)

where (t2, cs2) = parse cs ′

...
(tm, csm) = parse csm−1

= {pretty ′, ++}
approx (Succ n) (’L’ : pretty ′ t2 ++ . . . ++ pretty ′ tm ++ csm)

where (t2, cs2) = parse cs ′

...
(tm, csm) = parse csm−1

= {approx}
’L’ : approx n (pretty ′ t2 ++ . . . ++ pretty ′ tm ++ csm)

where (t2, cs2) = parse cs ′

...
(tm, csm) = parse csm−1

=
˘
ppm−1, m ≥ 1

¯
’L’ : approx n (ppm−1 cs ′)

v {Inductive hypothesis, monotonicity}
’L’ : approx n (id cs ′)

= {id , approx}
approx (Succ n) (’L’ : cs ′)

Last case, head cs = ’B’, i.e. cs = ’B’ : cs ′ for some cs ′ :: String :

approx (Succ n) (ppm (’B’ : cs ′))
= {ppm, m ≥ 1}
approx (Succ n) (pretty ′ t1 ++ pretty ′ t2 ++ . . . ++ pretty ′ tm ++ csm)

where (t1, cs1) = parse (’B’ : cs ′)
(t2, cs2) = parse cs1

...
(tm, csm) = parse csm−1

= {parse, where}
approx (Succ n) (pretty ′ (B l r) ++ pretty ′ t2 ++ . . . ++ pretty ′ tm ++ csm)

where (l , ls) = parse cs ′

(r , rs) = parse ls
(t2, cs2) = parse rs

...



...
(tm, csm) = parse csm−1

= {pretty ′, ++, ++ associative}
approx (Succ n)

(’B’ : pretty ′ l ++ pretty ′ r ++ pretty ′ t2 ++ . . . ++ pretty ′ tm ++ csm)
where (l , ls) = parse cs ′

(r , rs) = parse ls
(t2, cs2) = parse rs

...
(tm, csm) = parse csm−1

= {approx}
’B’ : approx n (pretty ′ l ++ pretty ′ r ++ pretty ′ t2 ++ . . . ++ pretty ′ tm ++ csm)

where (l , ls) = parse cs ′

(r , rs) = parse ls
(t2, cs2) = parse rs

...
(tm, csm) = parse csm−1

=
˘
ppm+1

¯
’B’ : approx n (ppm+1 cs ′)

v {Inductive hypothesis, monotonicity}
’B’ : approx n (id cs ′)

= {id , approx}
approx (Succ n) (’B’ : cs ′)

Hence we have yet again proved (2), this time using the approximation lemma.

Parse after Pretty. Let us now turn to (1’). We want to verify that parse ◦
pretty = strictify :: (T ,String) → (T ,String) holds. This can be done using the
approximation lemma as given in equivalence (3). To ease the presentation we
will use the following helper function:

approxP :: Nat → (T , a) → (T , a)
approxP n (t , a) = (approx n t , a)

Using this function we can formulate the approximation lemma as

p1 = p2 iff ∀n ∈ Natfin . approxP n p1 = approxP n p2 (13)

for arbitrary pairs p1 , p2 :: (T , ψ), where ψ is an arbitrary type. In our case
ψ = String , p1 = (parse ◦ pretty) p and p2 = strictify p for an arbitrary pair
p :: (T ,String).

The proof proceeds by induction over n as usual; and as usual we first have
to observe that parse ◦ pretty and strictify are both distinct from ⊥. The case
n = Zero is trivial. Now assume that we have proved approxP n ((parse ◦
pretty) p) = approxP n (strictify p) for some n ∈ Natfin and all p :: (T ,String).
(All p since we can change the order of the universal quantifiers like we did to
arrive at inequality (12).) We prove the corresponding statement for Succ n by



case analysis on p. All cases except for the one where p = (B l r , cs) for arbitrary
subtrees l , r :: T and an arbitrary cs :: String are straightforward and omitted,
so we go directly to the last case:

approxP (Succ n) ((parse ◦ pretty) (B l r , cs))
= {◦, pretty , pretty ′, ++, ++ associative}
approxP (Succ n) (parse (’B’ : pretty ′ l ++ pretty ′ r ++ cs))

= {parse, pretty , ◦}
approxP (Succ n) (B l ′ r ′, cs ′′)

where (l ′, cs ′) = (parse ◦ pretty) (l , pretty ′ r ++ cs)
(r ′, cs ′′) = parse cs ′

= {approxP , approx}
(B (approx n l ′) (approx n r ′), cs ′′)

where (l ′, cs ′) = (parse ◦ pretty) (l , pretty ′ r ++ cs)
(r ′, cs ′′) = parse cs ′

= {Push approx n through the pairs, turning it into approxP n}
(B l ′ r ′, cs ′′)

where (l ′, cs ′) = approxP n ((parse ◦ pretty) (l , pretty ′ r ++ cs))
(r ′, cs ′′) = approxP n (parse cs ′)

= {Inductive hypothesis}
(B l ′ r ′, cs ′′)

where (l ′, cs ′) = approxP n (strictify (l , pretty ′ r ++ cs))
(r ′, cs ′′) = approxP n (parse cs ′)

= {strictify}
(B l ′ r ′, cs ′′)

where (l ′, cs ′) = approxP n (l ‘seq ‘ (t ′, tTotal ‘seq ‘ pretty ′ r ++ cs))
(t ′, tTotal) = strictify ′ l
(r ′, cs ′′) = approxP n (parse cs ′)

The proof proceeds by case analysis on l . We omit the cases l = ⊥ and l = L
and go to the last case, l = B l1 r1 for arbitrary subtrees l1, r1 :: T :

(B l ′ r ′, cs ′′)
where
(l ′, cs ′) = approxP n (B l1 r1 ‘seq ‘ (t ′, tTotal ‘seq ‘ pretty ′ r ++ cs))
(t ′, tTotal) = strictify ′ (B l1 r1)
(r ′, cs ′′) = approxP n (parse cs ′)

= {seq , strictify ′, where}
(B l ′ r ′, cs ′′)

where
(l ′, cs ′) = approxP n (B l ′1 (lTotal ‘seq ‘ r ′1),

(lTotal ‘seq ‘ rTotal) ‘seq ‘ pretty ′ r ++ cs)
(l ′1, lTotal) = strictify ′ l1
(r ′1, rTotal) = strictify ′ r1
(r ′, cs ′′) = approxP n (parse cs ′)

= {approxP , where}
(B (approx n (B l ′1 (lTotal ‘seq ‘ r ′1))) r ′, cs ′′)

where
(l ′1, lTotal) = strictify ′ l1
(r ′1, rTotal) = strictify ′ r1
(r ′, cs ′′) = approxP n (parse ((lTotal ‘seq ‘ rTotal) ‘seq ‘ pretty ′ r ++ cs))



Now we have two cases, depending on whether lTotal ‘seq ‘ rTotal , i.e. snd
(strictify ′ l1) ‘seq ‘ snd (strictify ′ r1), equals ⊥ or not. We omit the case where
the equality holds and concentrate on the case where lTotal ‘seq ‘rTotal = () 6= ⊥:

(B (approx n (B l ′1 (lTotal ‘seq ‘ r ′1))) r ′, cs ′′)
where (l ′1, lTotal) = strictify ′ l1

(r ′1, rTotal) = strictify ′ r1
(r ′, cs ′′) = approxP n (parse (() ‘seq ‘ pretty ′ r ++ cs))

= {seq , pretty , ◦}
(B (approx n (B l ′1 (lTotal ‘seq ‘ r ′1))) r ′, cs ′′)

where (l ′1, lTotal) = strictify ′ l1
(r ′1, rTotal) = strictify ′ r1
(r ′, cs ′′) = approxP n ((parse ◦ pretty) (r , cs))

= {Inductive hypothesis}
(B (approx n (B l ′1 (lTotal ‘seq ‘ r ′1))) r ′, cs ′′)

where (l ′1, lTotal) = strictify ′ l1
(r ′1, rTotal) = strictify ′ r1
(r ′, cs ′′) = approxP n (strictify (r , cs))

= {Push approxP n through the pair, turning it into approx n}
(B (approx n (B l ′1 (lTotal ‘seq ‘ r ′1))) (approx n r ′), cs ′′)

where (l ′1, lTotal) = strictify ′ l1
(r ′1, rTotal) = strictify ′ r1
(r ′, cs ′′) = strictify (r , cs)

= {approx , approxP}
approxP (Succ n) (B (B l ′1 (lTotal ‘seq ‘ r ′1)) r ′, cs ′′)

where (l ′1, lTotal) = strictify ′ l1
(r ′1, rTotal) = strictify ′ r1
(r ′, cs ′′) = strictify (r , cs)

The rest of the proof consists of transforming the expression above to approxP
(Succ n) (strictify (B (B l1 r1) r , cs)). This is relatively straightforward and
omitted. Thus we have, yet again, proved (1’).

9 Discussion and Future Work

In this paper we have investigated how different verification methods can handle
partial and infinite values in a simple case study about data conversion. We have
used random testing, fixpoint induction and the approximation lemma.

Using isBottom and approx for testing in the presence of partial and infinite
values is not fool proof but works well in practice. The approach is not that
original; testing using isBottom and take is (indirectly) mentioned already in
the original QuickCheck paper [2]. However, testing using approx has probably
not been done before. Furthermore, the functionality of =̂ and v̂ has not been
provided by any (widespread) library.

The two methods used for proving the properties (1’) and (2) have different
qualities. Fixpoint induction required us to rewrite both the functions and the
properties. Furthermore one property did not hold for the base case, so it had to



be rewritten (7’), and proving the base case required some tedious but straight-
forward work. On the other hand, once the initial work had been completed
the “actual proofs” were comparatively short. The corresponding “actual proofs”
were longer when using the approximation lemma. The reason for this is proba-
bly that the approximation lemma requires that the function approx is “pushed”
inside the expressions to make it possible to apply the inductive hypothesis. For
fixpoint induction that is not necessary. For instance, when proving (1’) using the
approximation lemma we had to go one level further down in the tree when per-
forming case analysis, than in the corresponding proof using fixpoint induction.
This was in order to be able to use the inductive hypothesis.

Nevertheless, the “actual proofs” are not really what is important. They
mostly consist of performing a case analysis, evaluating both sides of the (in-)
equality being proved as far as possible and then, if the proof is not finished
yet, choosing a new expression to perform case analysis on. The most important
part is really finding the right inductive hypothesis. (Choosing the right expres-
sion for case analysis is also important, but easier.) Finding the right inductive
hypothesis was easier when using fixpoint induction than when using the ap-
proximation lemma. Take the proofs of (2), for instance. When using fixpoint
induction almost no thought was needed to come up with the inductive hypoth-
esis, whereas when using the approximation lemma we had to come up with the
complex hypothesis based on property (10), the one involving ppm. The reason
was the same as above; approx has to be in the right position. It is of course
possible that easier proofs exist.

It is also possible that there are other proof methods which work better
than the ones used here. Coinduction and fusion, two other methods mentioned
in Gibbons’ and Hutton’s tutorial [4], might belong to that category. We have
made some attempts at using unfold fusion. Due to the nature of the programs
the standard fusion method seems inapplicable, though; a monadic variant is a
better fit. The programs can be transformed into monadic variants (which of
course carries extra proof obligations). We have not yet figured out where to
go from there, though. For instance, the monadic anamorphism fusion law [10,
Equation (17)] only applies to a restrictive class of monads, and our “monad”
does not even satisfy all the monad laws (compare Sect. 3).

Above we have compared different proof techniques in the case where we
allow infinite and partial input. Let us now reflect on whether one should consider
anything but finite, total values. The proofs of (1’) and (2) valid for all inputs
were considerably longer than the ones for (1) and (2) limited to finite (and in
one case total) input, especially if one takes into account all work involved in
rewriting the properties and programs. It is not hard to see why people often
ignore partial and infinite input; handling it does seem to require nontrivial
amounts of extra work.

However, as argued in Sect. 1 we often need to reason about infinite values.
Furthermore, in reality, bottoms do occur; error is used, cases are left out from
case expressions, and sometimes functions do not reach a weak head normal form
even if they are applied to total input (for instance we have reverse [1 . . ] = ⊥).



Another reason for including partial values is that in our setting of equational
reasoning it is easier to use a known identity if the identity is valid without
a precondition stating that the input has to be total. Of course, proving the
identity without this precondition is only meaningful if the extra work involved
is less than the accumulated work needed to verify the precondition each time
the identity is used. This extra work may not amount to very much, though.
Even if we were to ignore bottoms, we would still sometimes need to handle
infinite values, so we would have to use methods like those used in this text. In
this case the marginal cost for also including bottoms would be small.

Another approach is to settle for approximate results by e.g. assuming that
λx → ⊥ is ⊥ when reasoning about programs. These results would be practically
useful; we might get some overly conservative results if we happened to evaluate
seq (λx → ⊥), but nothing worse would happen. On the other hand, many
of the caveats mentioned in Sect. 3 would vanish. Furthermore most people
tend to ignore these issues when doing ordinary programming, so in a sense an
approximate semantics is already in use. The details of an approximate semantics
for Haskell still need to be worked out, though. We believe that an approach
like this will make it easier to scale up the methods used in this text to larger
programs.

A QuickCheck Generators

The QuickCheck generators used in this text are defined as follows:

tree :: Gen T
tree = frequency [(6, liftM2 B tree tree),

(2, return L),
(1, return ⊥)]

string :: Gen String
string = frequency [(1, bottomString),

(1,finiteString),
(1, infiniteString),
(3, treeString)]

where
bottomString = liftM2 approx arbitrary infiniteString
finiteString = liftM2 (take ◦ abs) arbitrary infiniteString
infiniteString = liftM2 (:) char infiniteString
treeString = tree >>= return ◦ pretty ′

char :: Gen Char
char = frequency [(10, return ’B’),

(10, return ’L’),
(1, return ’?’),
(1, return ⊥)]

pair :: Gen (T ,String)
pair = frequency [(50, liftM2 (, ) tree string),

(1, return ⊥)]



A straightforward Arbitrary instance for Nat (yielding only total, finite values)
is also required.

The generator tree is defined so that the generated trees have a probability
of 1

2 of being finite, and the finite trees have an expected depth of 2.2 We do not
generate any total, infinite trees. The reason is that some of the tests above do
not terminate for such trees, as shown in Sect. 5.

To get a good mix of finite and infinite partial strings the string generator
is split up into four cases. The last case ensures that some strings that actually
represent trees are also included. It would not be a problem to include total,
infinite strings, but we do not want to complicate the definitions above too
much, so they are also omitted.

Finally the pair generator constructs pairs using tree and string , forcing some
pairs to be ⊥.

By using collect we have observed that the actual distributions of generated
values correspond to our expectations.

2 Assuming that QuickCheck uses a random number generator that yields independent
values from a uniform distribution.


